
''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie
'''''''''''''''''''''''''''''''''''''''''''' v 8, Aug. 2008

'   Purpose

'   Installation 

'   The MacroMorsels

'     Data input & output MacroMorsels

'       InputBoxForCell          

'       InputBoxForRange         

'       InputBoxForArray         

'       OutputAnArray1 

'       OutputAnArray2 

'       OutputAnArray3 

'       PreventCellOverwrite             

'       PreventColumnOverwrite             

'       Rounding

'       ImportingData

'       UseAnExcelFunction1        

'       InvasiveComputing

'       CentralDifferencing2

'       Counting Pennies

'     MacroMorsels for spreadsheet and macro management 

'
  ScanSymbolCode

'
  UseSymbolCode

'       ControlANumericalDisplay 

'       DeconstructACellAddress

'       DeconstructAnArrayAddress

'       ReconstituteAnEquation1

'       ReconstituteAnEquation2

'       MoveAndResizeARange

'       InhibitScreenUpdating    

'       KeepUserInformed         

'       AttachCellComments

'       MakeTestToolbar         

'       ErrorPrevention         

'       ErrorTrapping(1)         

' PURPOSE
' MacroMorsels are miniature macros that can be used as

' modules in building Excel macros. Each illustrates just 

' one feature of Excel VBA macro writing. They are complete

' macros, properly dimensioned, with explanatory comments,

' ready to be used. They show their action directly on the   

' spreadsheet, or display their results in either the   

' Immediate Window or a message box.

' The primary focus of these MacroMorsels is on the inter-

' play between VBA and the spreadsheet, because that is 

' where you will initially spend most time debugging macro 

' code. 

' As in any computer language, there are usually many 

' different ways to achieve a given objective, yet in many 

' cases only one is shown here. The purpose is to get you

' started with something that works, rather than getting

' you confused with a number of choices. After you find one 

' route, you can experiment with alternatives.

' Most MacroMorsels are organized as follows:

'     Macro name              Sub name() 

'     Explanatory comments    (if necessary)

'     Dimensioning

'     Operating code

'     Testing code    

'
Conclusions             (where not obvious)


'
Notes             

(when needed)


'     Closing line            End Sub

' The MacroMorsels have all been dimensioned to work with

' (but don't require you to use) Option Explicit. Essential 

' parts of their operating codes are often bolded and/or 

' colored in this Word version for extra emphasis, but such 

' text effects will of course disappear when these 

' Macromorsels are inserted in a VBEditor module. 

' For the sake of simplicity, the MacroMorsels take their

' input either through highlighting of a spreadsheet cell  

' or cell block, or through a simple input box. 

' Results that would not show on the spreadsheet are 

' displayed in a message box, and/or in the Immediate

' Window. The latter can be opened from within the VBEditor 

' with either Ctrl+g or View > Immediate Window, and sized

' by moving its upper edge when it shows two parallel 

' horizontal lines. The Immediate Window is shown in the 

' bottom right corner of the VB Editor module. 

' Before incorporating the code of a MacroMorsel into your 

' own macro, first test it, as is, in order to familiarize 

' yourself with its operation. If it does what you want it 

' to do, copy it (without its first and last line) into 

' your own macro code, adjust its parameter names to fit 

' your own naming convention, test it in its new 

' environment, perhaps clean it up some more, and use it.

' Because Excel 2007 has not changed its VBA, these macros  

' can be used unchanged in this latest version of Excel. 

' This collection is a work in progress, and your 

' suggestions are welcome. Please send them by email to 

' rdelevie@bowdoin.edu

' NOMENCLATURE
' In order to make these MacroMorsels as transparent as 

' possible, we often use the following prefixes to 

' identify the nature of the parameters used:

'   a:  Address

'   c:  Column

'   f:  Formula

'   r:  Row

'   rg: Range

'   v:  Value

' We use capitals for readability, and to exploit the 

' typo-catching feature of Option Explicit. It also yields  

' more compact parameter names than using underscores.

' COPYRIGHT and ABSENCE OF WARRANTY 
' These MacroMorsels are copyrighted, and are licenced  

' under the GNU General Public License. In publications  

' that use them, please acknowledge this web site as 

' their source, so that other potential users can learn   

' about and benefit from them.  

' These MacroMorsels are made available without any 

' warranty whatsoever. Prospective users are advised to 

' check the performance of these MacroMorsels on their  

' own computer to make sure that they work properly in 

' this setting. Some operations may not work in early

' versions of Excel; some features used here may not be 

' available in your particular version of Excel. Some 

' conventions and commands may be different in non-US 

' versions.

' INSTALLATION 
' First, store this file somewhere on your computer, and 

' perhaps print a copy of this document for easy reference. 

' Select this entire set of MacroMorsels, including these 

' comments) with Edit -> Select All, copy it to the clip-

' board (e.g., with Ctrl+c), and subsequently open Excel.

' Go to the VBEditor (e.g., with Alt+F11 or, on the Mac,

' with Opt+F11), and in its menu select Insert -> Module. 

' Then paste (Ctrl+v) the MacroMorsels into that module. 

' Press Ctrl+g to open the Immediate Window. This is where

' many output results of the MacroMorsels will be shown.

' The Immediate Window will stay open until you close it

' with the X closing button in its right top corner.

' Return to the spreadsheet with Alt+F11 (Mac: Opt+F11).

' With the above you will have access to all MacroMorsels

' from the spreadsheet via Alt+F8 (Mac: Opt+F8), and from

' the specific MacroMorsel in the VBEditor with F5. 

' PRELIMINARIES 
' Before testing a particular MacroMorsel, read the few 

' associated comment lines. This is essential, because

' different MacroMorsels have different input requirements.

' You can do this in the printed copy or, if you prefer, 

' in the VBEditor module where the MacroMorsels are stored.  

' For the latter, click Alt+F8 (Mac: Opt+F8), click once 

' on the macro name to highlight it (a double click would

' call it, i.e., would start its operation), then click on

' Edit. After reading the comments, you can activate the 

' MacroMorsel (as long as the pointer remains inside that 

' particular MacroMorsel) with the function key F5. 

' Alternatively you can return to the spreadsheet with 

' Alt+F11 (Mac: Opt+F11), and then use Alt+F8 (Mac: Opt+F8) 

' to select the MacroMorsel by double-clicking on it. 

' Make sure that the Immediate Window is visible at the 

' bottom of the VBEditor module. If not, press Ctrl+g. 

' Results of the MacroMorsels may show in that Immediate 

' Window. 

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'            Data input & output MacroMorsels

    '
'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ReadActiveCell()
' Purpose: to read (i.e., copy) the contents of a single 

' spreadsheet cell into a macro.

' The cell must be activated (highlighted) before the macro

' is called. In case you do not need all three pieces of

' information (address, formula, value), just delete the

' superfluous one(s). Check the answers in the Immediate

' Window.

Dim vX As Variant



' dimension
Dim aX As String

Dim fX

aX = Selection.Address


' read
fX = Selection.Formula


vX = Selection.Value


Debug.Print "Address: " & aX

' validate
Debug.Print "Formula: " & fX

Debug.Print "Value:   " & vX

' Note: this macro actually reads a single-cell range.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ReadActiveCellBlock()
' Purpose: to read (i.e., copy) the contents of a single, 

' rectangular block of cells into a macro. This macro also 

' works for a single row or a single column as input, but 

' NOT for a single cell. The reason why is mentioned below.

' The array must be activated (highlighted) BEFORE the

' macro is called. In case you do not need all three

' pieces of information (address, formula, value), just

' delete the superfluous one(s).

' For fX and vX, the order is important: first count rows

' and columns, then redimension fX and vX accordingly, and 

' finally read them.

Dim c As Integer, cMax As Integer

Dim r As Integer, rMax As Integer

Dim fX As Variant, vX As Variant

Dim aX As String

cMax = Selection.Columns.Count     ' count
rMax = Selection.Rows.Count

If cMax = 1 And rMax = 1 Then      ' check input format
  MsgBox "An array must contain more than one cell"

  End

End If

ReDim fX(1 To rMax, 1 To cMax)     ' redimension
ReDim vX(1 To rMax, 1 To cMax)

aX = Selection.Address             ' read
fX = Selection.Formula
vX = Selection.Value

Debug.Print "Array Address: " & aX ' validate
For c = 1 To cMax

  For r = 1 To rMax

    Debug.Print "Formula(" & r & "," & c & "): " & fX(r, c)

    Debug.Print "Value(" & r & "," & c & "):   " & vX(r, c)

  Next r

Next c

' Note: this macro reads a multi-cell range, then

' extracts from this range the information for two arrays,

' one for the formulas, the other for the values. In the

' current example, this facilitates displaying these

' quantities in the Immediate Window.

' The restriction to a multi-cell input block stems from

' the fact that an array must contain multiple components,

' quite distinct from a range.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub InputBoxForNumber()

' If all you need is to input a number, here is a simple

' way to do so. No cell needs be activated (highlighted)

' before the macro is called. Just call the macro, enter

' the number in the input box, and press OK (on the input

' box) or Enter (on the keyboard). The first item within

' brackets after InputBox is the prompt i.e., text shown

' inside the input box to remind the user what kind of

' response is expected.

Dim DataInput As Double

DataInput = InputBox("Enter a number:")
Debug.Print "Box Input = " & DataInput

' There are several available options, of which two are 

' illustrated below, such as a title (shown in the blue 

' title strip at the top of the input box), and a default

' value (to be used as input if one hits the OK or Enter

' key without entering a number). In that case the

' instruction reads instead

' DataInput = InputBox("Enter a number", _

  "InputBoxForNumber", 13)

' Note that the order of the items listed within brackets 

' defines their function. In case one wants, e.g., to use

' a default value but no title, the instruction must

' retain the two commas preceding the default value, and 

' merely delete the title, and should therefore read

' DataInput = InputBox("Enter a number", , 13)

' Try these alternatives for DataInput.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub InputBoxForCell()

' Purpose: to read (i.e., copy) the contents of a single 

' spreadsheet cell, as a range, into a macro, using an 

' input box rather than highlighting.

' The cell need not be activated (highlighted) before the

' macro is called. Instead, the cell address must be typed

' in (or dragged into) the input box.

Dim rgX As Range        
Set rgX = Application.InputBox(Prompt:="InputCell: ", _

  Type:=8)

rgX.Select

Debug.Print "Cell address: " & rgX.Address

Debug.Print "Cell formula: " & rgX.Formula

Debug.Print "Cell value:   " & rgX.Value

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub InputBoxForRange()

' Purpose: to read (i.e., copy) the contents of a (single-

' or multi-cell) range from the spreadsheet into a macro,

' by using an input box. The contents include the range 

‘ address, and the formulas and values of the individual 

‘ range elements.

' Call the macro, highlight the array (so that it shows in

' the input box window) or type the array address, then

' click OK or Enter.
Dim c As Long, cMax As Long ' just in case these exceed 32K
Dim r As Long, rMax As Long 

Dim vA As Double

Dim rgA As Range

Dim fA 

Set rgA = Application.InputBox(Prompt:="Range", _

  Title:="Test", Type:=8)
rgA.Select

cMax = Selection.Columns.Count

rMax = Selection.Rows.Count

Debug.Print "Address: " & rgA.Address
' validate
Debug.Print ""

Debug.Print " r", " c", "Formula", "Value"

Debug.Print ""

For r = 1 To rMax

  For c = 1 To cMax

    fA = rgA.Item(r, c).Formula

    vA = rgA.Item(r, c).Value

    Debug.Print r, c, fA, vA

  Next c

Next r

' Because a multi-cell range has no singular formula

' or value, we first limit fA and vA to a specific range

' item before we can find its formula or value.

' The term Item is optional; you can also use, e.g.,

'       fA = rgA(r, c).Formula

' while in many versions you may get away with, e.g.,

'       fA = rgA.Formula(r, c)

' but at the risk of malfunctioning in a different version.

' You can use the same approach to extract other range

' information, such as font size, color, etc. of one or 

' more range cells, as with

'       Dim ColorA

'       ColorA = rgA(r, c).Font.ColorIndex

' in which case you may want to add ColorA to the list of

' items displayed in the Immediate Window.

' In general it is faster and more convenient to transfer

' the range information into arrays (as in the next

' MacroMorsel) if you want that information for all cells

' in a range; the approach used here is simpler when that

' information is required for only one or a few cells, 

' i.e., for specific values of c and r.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub InputBoxForArray()

' Purpose: to read (i.e., copy) the contents of a single 

' spreadsheet array (in a rectangular block of cells)  

' into a macro. The approach is to read the range into 

' the macro, and then extract the needed array(s) from it.   

' Use Type:=8 for a rather general input.

' The array need not be activated (highlighted) before the

' macro is called. Use Type:=8 for a rather general input.

Dim c As Long, cMax As Long

Dim r As Long, rMax As Long 

Dim vX As Variant

Dim rgX As Range

Set rgX = Application.InputBox(Prompt:="InputArray: ", _

  Type:=8)

rgX.Select

cMax = rgX.Columns.Count

rMax = rgX.Rows.Count

ReDim vX(1 To rMax, 1 To cMax)

vX = Selection.Value


' extract vX 

For c = 1 To cMax



' validate 

  For r = 1 To rMax

    Debug.Print "Value(" & r & "," & c & "):   " & vX(r, c)

  Next r

Next c

' Instead, or in addition, you can ask for the Formula

' and/or Address of the various array elements.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub OutputASingleValue1()

' This one-line MacroMorsel illustrates writing a macro

' value in an activated (or highlighted) spreadsheet cell.

Selection.Value = 7654321

' You can also write text this way, by using double

' quotation marks, as in

'       Selection.Value = "The answer is ready."

' Try it.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub OutputASingleValue2()

' This variation uses a named variable, here called N,

' to which a value has been assigned

       Dim N As Double

       N = 2345.67

       Selection.Value = "N = " & N

' and combines text and data. The same happens in

'       Dim A As Double, N As Double

'       N = 2345.67

'       A = Sqr(N)

'       Selection.Value = "N = " & N & ", A = " & A

' but such mixed messages should in general be avoided,

' since such combinations of text plus data are strings

' that cannot be used as such in subsequent calculations.

' It is therefore preferable to output the data and their

' labels separately, as in OutputSeveralValues

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub OutputSeveralValues()

' Here is a better way to output several values onto the

' spreadsheet, by putting label and each value in its own,

' separate spreadsheet cell:

       Dim A As Double, N As Double

       N = 2345.67

       A = Sqr(N)

       Selection.Value = "N ="

       Selection.Offset(0, 1).Select

       Selection.Value = N

       Selection.Offset(1, -1).Select

       Selection.Value = "A ="

       Selection.Offset(0, 1).Select

       Selection.Value = A

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub OutputAnArray()

' This MacroMorsel illustrates how an array computed in a

' macro can be placed on the spreadsheet, by first reading 

' it into a range, communicating that range with the spread- 

' sheet, and then extracting the values. For array we will

' first compute a table of Student t values, used in con-

' verting standard deviations into "confidence" measures.

' Before calling this macro, activate the cell you want to

' to become the top left-hand corner cell of the array,

' i.e., click on that cell.

Dim c As Integer, cMax As Integer

Dim r As Integer, rMax As Integer

cMax = 4: rMax = 15

ReDim t(1 To rMax, 1 To cMax) As Double

Dim a(1 To 4) As Double

Dim rgA As Range

a(1) = 0.1: a(2) = 0.05: a(3) = 0.01: a(4) = 0.001

' Compute the table of Student t values

' Here, a(c) specifies the probability (1 minus the confi-

' dence interval), and r the number of degrees of freedom.

For r = 1 To rMax

  For c = 1 To cMax

    t(r, c) = Application.TInv(a(c), r)

  Next c

Next r

' Use a range to write the array values on the spreadsheet

' Note: the left-top cell of the array will be placed at

' the current location of the active cell. WARNING: Unless 

' you have carefully positioned the pointer there, and have  

' an idea of the size of the array, this risks overwriting 

' useful data.

Set rgA = ActiveCell.Range(Cells(1, 1), Cells(rMax, cMax))

rgA.Value = t

' Note: An easier method is available if the subroutine

' already contains an array, especially one of the desired

' dimension, as illustrated in the CentralDifferencing2

' MacroMorsel. In that example, the range rgA was used to 

' dimension the array Fderiv with

'              Fderiv = rgA.Value

' After computing the individual terms Fderiv(i,1) the

' output was then placed to the immediate right of rgA with

'              rgA.Select

'              rgA.Select

'              cA = Selection.Columns.Count ' If needed

'              rgA.Select

'              Selection.Offset(0, cA).Select

'              Selection.Value = Fderiv

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub OutputAnArray2()

' This MacroMorsel extends the earlier one by including

' a heading, Student t, and the column labels 0.1, 0.05,

' 0.01, and 0.001.

' Before calling this macro, activate the cell you want to

' to become the top left-hand corner cell of the array,

' i.e., click on that cell.

Dim c As Integer, cMax As Integer

Dim r As Integer, rMax As Integer

cMax = 4: rMax = 15

ReDim t(1 To rMax + 3, 1 To cMax)

Dim A(1 To 4) As Double

Dim rgA As Range

A(1) = 0.1: A(2) = 0.05: A(3) = 0.01: A(4) = 0.001

' Compute the table of Student t values

' Here, a(c) specifies the probability (1 minus the confi-

' dence interval), and r the number of degrees of freedom.

t(1, 1) = "Student t"

t(2, 1) = A(1)

t(2, 2) = A(2)

t(2, 3) = A(3)

t(2, 4) = A(4)

For r = 4 To rMax + 3
  For c = 1 To cMax

    t(r, c) = Application.TInv(A(c), r - 3)

  Next c

Next r

' Note the following changes to make this possible:

' (1) t is now redimensioned with three more rows, but

'     without specifying it As Double because t(1, 1)

'     now contains text;

' (2) the For ... Next loop starts counting at 4, and

'     the value of t(r, c) is now computed for r - 3; and

' (3) the output statement includes three more rows.

' More work, but also a more user-friendly output.

Set rgA = ActiveCell.Range(Cells(1, 1), _

  Cells(rMax + 3, cMax))

rgA.Value = t

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub OutputAnArray3()

' This MacroMorsel completes the labeling by adding a

' column with the number of degrees of freedom to the

' left of the Student t values.

' Before calling this macro, activate the cell you want to

' to become the top left-hand corner cell of the array,

' i.e., click on that cell.

Dim c As Integer, cMax As Integer

Dim r As Integer, rMax As Integer

cMax = 4: rMax = 15

ReDim t(1 To rMax + 3, 1 To cMax + 2)

Dim A(1 To 4) As Double

Dim rgA As Range

A(1) = 0.1: A(2) = 0.05: A(3) = 0.01: A(4) = 0.001

' Compute the table of Student t values

' Here, a(c) specifies the probability (1 minus the confi-

' dence interval), and r the number of degrees of freedom.

t(1, 4) = "Student t"

t(2, 2) = "probab.="

t(2, 3) = A(1)

t(2, 4) = A(2)

t(2, 5) = A(3)

t(2, 6) = A(4)

t(3, 1) = "N-P="

For r = 4 To rMax + 3

  t(r, 1) = r - 3

Next r

For r = 4 To rMax + 3

  For c = 3 To cMax + 2
    t(r, c) = Application.TInv(A(c - 2), r - 3)

  Next c

Next r

' The following additional changes have been made:

' (1) redimensioning c to two more columns;

' (2) additional labels for t(2, 2) and t(3, 1);

' (3) adding a For ... Next loop to compute N-P:

' (4) increasing the c-values in the counter in the

'     second For ... Next loop by 2, with a correspon-

'     ding decrease in c in the expression for TInv; and

' (5) extending the c-range by 2 in the Set rgA command.

Set rgA = ActiveCell.Range(Cells(1, 1), _

  Cells(rMax + 3, cMax + 2))

rgA.Value = t

' Note that the actual calculation is still only one line

' in the second For ... Next loop, but that dimensioning

' and output formatting take up most of the macro.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub FromArrayValuesToRangeValues()

' Usually it is easier and faster to process data in a VBA

' array than in a VBA range. In many of the above examples,

' a spreadsheet range is converted into one or more arrays. 

' But if one needs to go the other way, to convert array 

' values into range values, this macro illustrates how it 

' can be done. Note the different ways used to specify a 

' range address. See what happens when the range address 

' is either taken too small or too large.

Dim i As Integer, j As Integer

Dim vX(1 To 2, 1 To 3) As Variant

Dim vY(1 To 5, 1 To 4) As Variant

' Define X array

vX(1, 1) = 3: vX(1, 2) = 5: vX(1, 3) = 7

vX(2, 1) = 4: vX(2, 2) = 6: vX(2, 3) = 9

' Define Y array

For i = 1 To 5

  For j = 1 To 4

    vY(i, j) = 4 + 5 * i - 6 * j

  Next j

Next i

' Convert array values into range values

Range("C17:E18").Value = vX

Range("C20", "F24").Value = vY

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub PreventCellOverwrite()

' In writing from a macro onto a spreadsheet, make sure 

' that you don’t overwrite essential information. This

' MacroMorsel scans the cell where information will be 

' placed, and if not empty asks whether it can overwrite 

' it with its result. If the answer is No it will provide 

' its result in a message box.  
' In this example, use a highlighted cell, with a number, 

' formula, text, or without anything in the target cell to 

' its right. Start the macro by highlighting a cell, then 

' call the macro. 

' Alternatively, make a column of such target cells, 
' modify the penultimate line in this macro to read

'         Selection.Offset(1, -1).Select

' highlight the cell to the left of the top target cell,

' and call the macro. You can now step through each cell 

' in turn.

Dim Answer, n As Integer, Result As Double

n = 1

Result = 7

Selection.Offset(0, 1).Select

If IsEmpty(Selection) Or Selection.Text = " " Then n = 0

If n = 1 Then Answer = MsgBox _

  ("Overwrite highlighted cell?", vbYesNo, "Overwrite?")

If n = 0 Or Answer = vbYes Then Selection = Result

If Answer = vbNo Then MsgBox "The output is " & Result

Selection.Offset(0, -1).Select

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub PreventColumnOverwrite()

' This MacroMorsel scans the area where information will 

' be placed, alerts you when it might contain valuable 

' data, and provides the option of writing the information 

' elsewhere. 
' In this example, we will use a highlighted block, and 

' then check whether there is anything written one column

' to the right of that block. If there is, the macro writes 

' its answer in the Immediate Window (from where you can

' copy it to another place on the spreadsheet).

' Start the macro by highlighting a cell block, and then

' call the macro. The output will appear in the Immediate

' Window.
Dim cN As Integer, i As Long, j As Integer

Dim n As Integer, rN As Long

Dim outputData As Variant

Dim Answer, inputData

n = 0

cN = Selection.Columns.Count    ' Determine block size
rN = Selection.Rows.Count

inputData = Selection           ' Read block
outputData = Selection

ReDim outputData(1 To rN, 1 To cN)  ' Generate a result
For i = 1 To rN

  outputData(i, cN) = Sqr(i)

Next i

Selection.Offset(0, 1).Select   ' Check if area is empty
For i = 1 To rN

  If IsEmpty(Selection(i, cN)) Or _

    Selection(i, cN).Text = " " Then

    n = n

  Else

    n = n + 1

  End If

Next i

If n > 0 Then Answer = MsgBox _

  ("Overwrite row to right of selected block?", vbYesNo)

If Answer = vbNo Then

  For i = 1 To rN

    Debug.Print outputData(i, cN)

  Next i

ElseIf Answer = vbYes Or n = 0 Then

  Selection = outputData

End If

Selection.Offset(0, -1).Select  ' Restore original input
Selection = inputData

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'               Data analysis MacroMorsels                '
'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Rounding()

' This MacroMorsel illustrates the various rounding

' rules used in Excel and VBA. It uses a short set of

' test data, and displays its results in the Immediate

' Window. Note that the VBA function Rnd generates a

' random number between 0 and 1. The prefix op is used 

' here to identify operating instructions without

' triggering them.

Dim i As Integer

Dim X(1 To 33) As Double, XX(1 To 33) As Double

Dim opRoundDown(1 To 33) As Double

Dim opRoundExcel(1 To 33) As Double

Dim opRoundVBA(1 To 33) As Double

Dim opRoundUp(1 To 33) As Double

For i = 1 To 33

  XX(i) = 400 - (i - 1) * 25

  X(i) = XX(i) / 10000

Next i

For i = 1 To 33

  opRoundUp(i) = Application.RoundUp(X(i), 2)

  opRoundExcel(i) = Application.Round(X(i), 2)

  opRoundVBA(i) = Round(X(i), 2)

  opRoundDown(i) = Application.RoundDown(X(i), 2)

Next i

Debug.Print " X", " RUp", " RXL", " RVB", " RDn"

Debug.Print ""

For i = 1 To 33

Debug.Print X(i), opRoundUp(i), opRoundExcel(i), _

  opRoundVBA(i), opRoundDown(i)

Next i

Debug.Print ""

' The somewhat unusual way X(i) is defined, with XX(i) as

' an auxiliary, is designed to obtain numbers X(i) without

' hidden round-off errors. In this respect, the number 0 

' is especially transparent.   

' Despite its name, the Excel function RoundUp(X) rounds

' away from zero, i.e., it rounds up for X > 0, but

' rounds down for X < 0.

' The Excel function Round(X) rounds towards the nearest

' integer, except for a 5 (not followed by any non-zero

' digits), which is rounded away from zero.

' Excel's RoundDown(X) rounds towards zero, again despite

' its misleading name: it rounds down for X > 0, and

' rounds up for X < 0.

' Only the VBA function Round(X) uses unbiased (statisti-

' cian's, banker's) rounding, by rounding to the nearest 

' integer, except to the nearest even integer for a 

' trailing 5 (possibly followed zeros), in which case 

' "nearest integer" is obviously ambiguous. 

' Unbiased rounding is the only rounding that yields 

' unbiased results, and is therefore the only type of 

' rounding that should be used in scientific data analysis.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ImportingData()

' The previous MacroMorsel, Rounding, generated a file

' in the Immediate Window. It is nicely formatted there,

' but when you copy it and then paste it into Excel you

' will get a mess, because numbers on a single row may

' all be bunched together in a single cell. That not only

' renders them unsightly, but also rather useless for

' subsequent spreadsheet computations.

' The correct way to handle this is by

' (1) highlighting the data in the Immediate Window,

' (2) copying them with Ctrl+c onto the Clipboard,

' (3) pressing Start in the lower left-hand corner of the

'     screen,

' (4) selecting All Programs > Accessories > Notepad,

' (5) pasting the file into Notepad with Ctrl+v,

' (6) saving this file (temporarily) on, say, the desktop

'     and give it a name, say IWDump,

' (7) returning to the spreadsheet and selecting the left

'     upper corner of where you want the file to appear,

' (8) pressing File > Open, selecting Look in: Desktop,

'     File Name: IWDump, Files of type: all files, Open.

' Try it with the Rounding file.

' And note that this MacroMorsel doesn't contain any

' computer commands, but only people instructions!

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Truncation()

' This MacroMorsel illustrates the various truncation rules

' used in Excel and VBA, as well as the behavior of the

' custom function cTrunc(X). It uses a short set of test

' data, and displays its results in the Immediate Window.

Dim i As Integer

Dim X(1 To 33) As Double

Dim opInt(1 To 33) As Double

Dim opCInt(1 To 33) As Double

X(1) = 4: X(2) = 3.75: X(3) = 3.5: X(4) = 3.25

For i = 5 To 33

  X(i) = X(i - 4) - 1

Next i

For i = 1 To 33

  opInt(i) = Int(X(i))

  opCInt(i) = CInt(X(i))

Next i

Debug.Print " X", "Int", "CInt"

Debug.Print ""

For i = 1 To 33

Debug.Print X(i), opInt(i), opCInt(i)

Next i

Debug.Print ""

' The VBA function Int(X) truncates X (i.e., removes the

' decimal point and everything to its right) when X is an

' integer or a non-negative non-integer. When X is both

' non-integer and negative, Int(X) truncates X-1. In other

' words, anything beyond the decimal point will get Int(X)

' to yield the next-smaller integer.

' VBA does not allow us to use the Excel function Int(X)

' in a macro, but it is readily verified on a spreadsheet

' that it acts the same as the VBA function.

' Excel has a good truncation function in Trunc(X), apart

' from the fact that Excel doesn't distinguish between the

' integers 0 and -0, so that Trunc(X) yields 0 for -1<X<1.

' At any rate, VBA is not allowed to use this spreadsheet

' function.

' The VBA function CInt(X) uses unbiased (statistician's,

' banker's) truncation, by rounding to the nearest integer,

' except to the nearest even integer for a trailing 5 not

' followed by non-zero digits, where "nearest integer" is

' obviously ambiguous. This, and the fully equivalent VBA

' function Round(X,0), are the only rational ways to

' truncate a numerical value to lead to unbiased results.

' Now transfer these data to the spreadsheet, as described

' in the above MacroMorsel, ImportingData.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub UseAnExcelFunction1()

' Excel has many more functions than VBA. Fortunately,

' one can invoke many Excel functions in VBA by using

' the preface Application. When in doubt, instead of 

' typing Application. followed by the function name, type 

' Application.WorksheetFunction. and scroll through the

' list of available functions appearing as a small menu

' on the screen just below where you are typing. Then
' double-click to select the function you want to use.

' Beyond this, the instruction WorksheetFunction. is 

‘ superfluous in the VBA instruction.

' For this example, place a positive integer in a cell

' and call the macro, which will compute the factorial 

' a = n! of the integer n

Dim n As Single

Dim a As Double

n = Selection.Value

a = Application.Fact(n) 

Debug.Print n, a

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub UseAnExcelFunction2()

' Purpose: this illustrates both how to read a spreadsheet 

' range, and how to use an Excel array function in a VBA

' macro. For the sake of variety, the answer is displayed

' in a message box.

Dim rgX As Range

Dim XMin

Set rgX = Application _

  .InputBox(Prompt:="Enter an array", Type:=8)
rgX.Select

XMin = Application.Min(rgX)

' Validation

MsgBox XMin

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub UseAnExcelFunction3()

' This example illustrates a possible danger: Excel and VBA

' functions of the same name not always, perform the same

' operation! For example, Excel's log yields the 10-based

' logarithm, so that log(10) = 1, whereas VBA's log gives

' the natural log, e.g., ' ln(10) = 2.30258...

' Place a positive number in a cell and call the macro

Dim a As Double, b As Double, vX As Double

vX = Selection.Value

a = Application.Log(vX)' the Excel function log is 10-based
b = Log(vX)            ' while the VBA function log is the

                       ' e-based, natural logarithm, in 

                       ' science typically denoted as ln
' Validation
Debug.Print "Excel log(" & vX & ") = " & a & Chr(10) & _

  "  VBA log(" & vX & ") = " & b

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub UseAnExcelFunction4()

' The number pi can also be called with straight brackets

Dim a As Double, b As Double

a = Application.Pi

b = [Pi()]
Debug.Print a, b

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub UseANestedExcelFunction()
' When one wants to use a nested Excel function in VBA,

' each Excel function must be preceeded by Application.

' Below we combine Index and Linest to generate the

' equivalent of the Excel function Index (LinEst(y-range,

' x-range,stats,const),row-index,column-index) to fit a

' straight line to seven adjacent data points, and to

' extract its slope a1 and the corresponding standard

' deviation s1.

' Make a column of seven x-values, and one of seven

' y-values. Then call the macro; the result will show in

' a message box as well as in the immediate window.

Dim a1 As Double, s1 As Double

Dim rgX As Range, rgY As Range

' Enter the seven x-values

Set rgX = Application.InputBox(Prompt:= _

  "Enter a column of seven x-values: ", Type:=8)

rgX.Select

' Enter the corresponding seven y-values

Set rgY = Application.InputBox(Prompt:= _

  "Enter the corresponding" & Chr(13) & _

  "column of seven y-values: ", Type:=8)

rgY.Select

a1 = Application.Index(Application.LinEst(rgY, rgX, 1, 1) _

  , 1, 1)

s1 = Application.Index(Application.LinEst(rgY, rgX, 1, 1) _

  , 2, 1)

MsgBox "The slope is " & a1 & " and its" & Chr(13) & _

  "standard deviation is " & s1

Debug.Print "a1 = " & a1 & ",   s1 = " & s1

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub UseAnExcelMatrixOperation1()

' This illustration reads a highlighted square array of

' numbers from the spreadsheet. It then applies the Excel

' instruction MInverse to these numbers, and writes the

' result back over the original matrix. Any formulas in

' the original data array are lost.

' The array needs to be activated (highlighted) before

' the macro is called.

Dim c As Integer, cMax As Integer

Dim r As Integer, rMax As Integer

Dim vX As Variant, vY As Variant

' (Optional) check that the matrix is square

cMax = Selection.Columns.Count

rMax = Selection.Rows.Count

If cMax <> rMax Then

  MsgBox "Array must be square."

  End

End If

' Read the highlighted array

vX = Selection.Value

' Validation: This macro is primarily validated by its

' spreadsheet action. Calling the same array twice

' should return the original matrix.

Debug.Print "The input values are:"

For c = 1 To cMax

  For r = 1 To rMax

    Debug.Print "vX(" & r & "," & c & "):   " & vX(r, c)

  Next r

Next c

' Matrix inversion

vY = Application.MInverse(vX)

' Validation

Debug.Print " After matrix inversion, the result is:"

For c = 1 To cMax

  For r = 1 To rMax

    Debug.Print "vY(" & r & "," & c & "):   " & vY(r, c)

  Next r

Next c

' Write the inverse matrix back onto the spreadsheet

Selection.Value = vY

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub UseAnExcelMatrixOperation2()

' This illustration does the same as its namesake #1, 

' except that it does not overwrite the original array, 

' but writes its result, the matrix inverse, below it. 

' In so doing, the macro preserves input formulas in the 

' original array. For simplicity the check on a square 

' input array, and the writing in the Immediate Window, 

' are left out.

' Activate (highlight) the array before calling the macro.

Dim rMax As Integer

Dim vX As Variant, vY As Variant

' Read the highlighted array

vX = Selection.Value

rMax = Selection.Rows.Count

' Invert the matrix

vY = Application.MInverse(vX)

' Move the active area

Selection.Offset(rMax + 1, 0).Select

' Write the inverse matrix back onto the spreadsheet

Selection.Value = vY

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub UseAnExcelMatrixOperation3()

' This illustration does the same as number2, except that

' it must write its result, the determinant of the matrix,

' in a resized area, which we have embellished somewhat.

' Activate (highlight) the array before calling the macro.

Dim rMax As Integer

Dim D As Double

Dim vX As Variant

' Read the highlighted array

vX = Selection.Value

rMax = Selection.Rows.Count

' Determine the determinant

D = Application.MDeterm(vX)

' Move and resize the active area

Selection.Offset(rMax + 1, 0).Select

Selection.Resize(1, 1).Select

' Label and write the determinant onto the spreadsheet

Selection.Value = "D ="

ActiveCell.HorizontalAlignment = xlRight

Selection.Offset(0, 1).Select

Selection.Value = D

ActiveCell.Font.ColorIndex = 3

End Sub
'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub UseAnExcelMatrixOperation4()

' For the multiplication of two matrices, M1 and M2, we

' will use input boxes. The matrix dimensions will be

' checked to make sure that the number of columns of M1

' equals the number of rows of M2. The result will be

' returned to a user- selected site on the spreadsheet.

' Note that you cannot multiply ranges, only data arrays.

' Excel matrix operations only deal with numbers.

' There is no need to activate (highlight) anything before

' calling the macro.

Dim c As Integer, cMax1 As Integer, cMax2 As Integer

Dim r As Integer, rMax1 As Integer, rMax2 As Integer

Dim vM1 As Variant, vM2 As Variant, vM3 As Variant

Dim rgM1 As Range, rgM2 As Range, rgM3 As Range

' Read matrix M1

Set rgM1 = Application.InputBox _

  (Prompt:="Enter matrix M1", Type:=8)

rgM1.Select

cMax1 = rgM1.Columns.Count

rMax1 = rgM1.Rows.Count

vM1 = Selection.Value

' Read matrix M2

Set rgM2 = Application.InputBox _

  (Prompt:="Enter matrix M2", Type:=8)

rgM2.Select

cMax2 = rgM2.Columns.Count

rMax2 = rgM2.Rows.Count

vM2 = Selection.Value

' Check that their dimensions are compatible with matrix

' multiplication M1 x M2, i.e., that cMax1 = rMax2

If cMax1 <> rMax2 Then

  MsgBox "The number of columns in M1 must be" & vbCr & _

  "  equal to the number of rows in M2."

  End

End If

' Select the location of the (top left corner of the)

' output matrix M3

Set rgM3 = Application.InputBox _

  (Prompt:="Enter the top left cell for the" & vbCr & _

           "result, the product matrix M3", Type:=8)

rgM3.Select

rgM3.Resize(rMax1, cMax2).Select

vM3 = Selection.Value

' Resize the output array

ReDim vM3(1 To rMax1, 1 To cMax2)

' Multiply the data arrays vM1 and vM2

vM3 = Application.MMult(vM1, vM2)

' Write the result onto the spreadsheet

Selection.Value = vM3

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub InvasiveComputing()

' This MacroMorsel illustrates how one can change a para-

' meter on the spreadsheet, and then observe and use the

' consequences of that change. In particular, we will find

' the first derivative of a function F(A) by changing A.

' In order to use this macro, place A in one spreadsheet

' cell, and in another cell the algebraic formula F that

' depends on A (and possibly on other factors, in which

' case we here compute the partial derivative). Note that

' we will only find the numerical value of the derivative

' of the function, at its current value. Incidentally, by

' using a fraction that can be expressed exactly as a

' binary number we optimize the precision of the answer.

' For starters, just to illustrate how this macro works,

' try A = 3 and F = 4 + 5A^2 = 49, for which Fderiv should

' be 5 x 2A = 30, or F = 7 - 6ln(A) = 0.408326268, which

' should yield Fderiv = -6 / A = -2. Then change the value

' of A and/or the formula for F, to suit your own fancy. 

' Incidentally, by using a fraction that can be expressed 

' exactly as a binary number, as done here, we sometimes 

' (depending on the function and on the parameter values) 

' can optimize the precision of the result. Try it by 

' replacing the value of Delta, 1/1048576, by the rather 

' similar 1/1000000. Still, with a Delta value of the order 

' of 10^7 we should count on losing about five significant 

' decimals in the derivative, which is about the best one 

' can do in general.
Dim A As Double, Amin As Double, Aplus As Double

Dim Delta As Double, F As Double, Fderiv As Double 

Dim Fmin As Double, Fplus As Double

Dim rgA As Range, rgF As Range

Delta = 1 / 1048576

' Read the values of A and F

Set rgF = Application.InputBox(Prompt:= _

  "The function is located in ", Type:=8)

rgF.Select

F = rgF.Value

Set rgA = Application.InputBox(Prompt:= _

  "The parameter a is located in ", Type:=8)

rgA.Select

A = rgA.Value

' Modify A to Aplus and read the corresponding Fplus

Aplus = A * (1 + Delta)

Selection.Value = Aplus

rgF.Select

Fplus = rgF.Value

' Modify A to Aminus and read the corresponding Fminus

Aminus = A * (1 - Delta)

rgA.Select

Selection.Value = Aminus

rgF.Select

Fminus = rgF.Value

' Clean up by restoring the initial value of A

rgA.Select

Selection.Value = A

' Compute and display the derivative

Fderiv = (Fplus - Fminus) / (2 * A * Delta)

MsgBox "The first derivative of the function is " & Fderiv

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub CentralDifferencing1()

' This macro illustrates central differencing of first,

' second, and third orders, for fixed Delta values. It

' computes the first derivative dF/da of the function F(a)

' with respect to the parameter a, where F and a are 

' located in single cells on the spreadsheet. The routine

' uses input boxes to ask for those locations.

' Apn is an abbreviation for A*(1+n*Delta), as in Ap4 =

' A*(1+4*Delta), where p denotes plus and n multiplies

' Delta. Likewise, Amn = A*(1-n*Delta), where m stands

' for minus, and the same applies to Fpn and Fmn.

' The output will be placed in three cells, in a column

' starting to the right of the function. Therefore,

' keep that space clear.

' If you use a test function with a known, exact value of

' its derivative, compute in the next column (starting

' four cells to the right of the function) the difference

' between the computed and exact value, in order to get

' an idea of the errors involved. Better yet, compute pE.

' Note that, for any given order, the errors still depend

' on the particular test function used, because there is

' no single, generally applicable value for Delta that 

' would be optimal for all test functions at that order of

' central differencing, but only a "ballpark" value of the

' approximately correct order of magnitude. 

Dim i As Integer

Dim Am4 As Double, Am2 As Double, Am1 As Double

Dim Ap1 As Double, Ap2 As Double, Ap4 As Double

Dim Delta(1 To 3) As Double, Fderiv(1 To 3) As Double

Dim Fm4 As Double, Fm2 As Double, Fm1 As Double

Dim Fp1 As Double, Fp2 As Double, Fp4 As Double

Dim vA As Double, vF As Double

Dim rgA As Range, rgF As Range

Dim fA

Delta(1) = 0.0000001

Delta(2) = 0.00002

Delta(3) = 0.004

Set rgF = Application.InputBox(Prompt:= _

  "The function F is located in ", Type:=8)

rgF.Select

vF = rgF.Value

Set rgA = Application.InputBox(Prompt:= _

  "The parameter a is located in ", Type:=8)

rgA.Select

vA = rgA.Value

fA = rgA.Formula

' Determine the derivative dF/da for the central

' differencing of first, second, and third order

For i = 1 To 3

' Modify vA and read the corresponding F-values

  Am4 = vA * (1 - 4 * Delta(i))

  Selection.Value = Am4

  rgF.Select

  Fm4 = rgF.Value

  Am2 = vA * (1 - 2 * Delta(i))

  rgA.Select

  Selection.Value = Am2

  rgF.Select

  Fm2 = rgF.Value

  Am1 = vA * (1 - Delta(i))

  rgA.Select

  Selection.Value = Am1

  rgF.Select

  Fm1 = rgF.Value

  Ap1 = vA * (1 + Delta(i))

  rgA.Select

  Selection.Value = Ap1

  rgF.Select

  Fp1 = rgF.Value

  Ap2 = vA * (1 + 2 * Delta(i))

  rgA.Select

  Selection.Value = Ap2

  rgF.Select

  Fp2 = rgF.Value

  Ap4 = vA * (1 + 4 * Delta(i))

  rgA.Select

  Selection.Value = Ap4

  rgF.Select

  Fp4 = rgF.Value

' Restore A

  rgA.Select

  Selection.Formula = fA

' Compute the derivative dF/da

  If i = 1 Then Fderiv(1) = (Fp1 - Fm1) _

    / (2 * Delta(i) * vA)

  If i = 2 Then Fderiv(2) = (Fm2 - Fp2 _

    + 8 * (Fp1 - Fm1)) / (12 * Delta(i) * vA)

  If i = 3 Then Fderiv(3) = (Fp4 - Fm4 + 40 * _

    (Fm2 - Fp2) + 256 * (Fp1 - Fm1)) _

    / (360 * Delta(i) * vA)

Next i

' Write the results to the spreadsheet

rgF.Select

Selection.Offset(0, 1).Select

Selection.Value = Fderiv(1)

Selection.Offset(1, 0).Select

Selection.Value = Fderiv(2)

Selection.Offset(1, 0).Select

Selection.Value = Fderiv(3)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub CentralDifferencing2()

' This MacroMorsel illustrates second order central

' differencing, used here to find the partial derivatives

' of a function of two or more parameters. (Second order

' central differencing is used here as a compromise: it is

' usually sufficiently accurate and, because it uses a

' smaller value of Delta, it is not quite as sensitive to

' steep functions as higher-order central differencing.

' The difference with CentralDifferencing1 lies in the way

' the range of parameter values is modified: one value at

' a time, using the Replace instruction.

' Put the input parameters in a single, contiguous column.

' The output will then be placed in a column right next to

' the input parameter values; keep that space clear.

' If you use a test function with exactly known values of

' its partial derivatives, compute in the next column

' (starting two cells to the right of the function) the

' difference between the computed and exact values, or the 

' corresponding pE-value, again in order to get a good idea 

' of the errors involved.

Dim i As Integer, rA As Integer

Dim Delta As Double, Fm2 As Double, Fm1 As Double

Dim Fp1 As Double, Fp2 As Double, vF As Double

Dim Am2 As Variant, Am1 As Variant, Ap1 As Variant

Dim Ap2 As Variant, vA As Variant, Fderiv As Variant

Dim rgA As Range, rgF As Range

Dim fA

Delta = 0.00002

' Input the parameters

Set rgA = Application.InputBox(Prompt:= _

  "The parameters are located in ", Type:=8)

rgA.Select

vA = rgA.Value

fA = rgA.Formula

rA = Selection.Rows.Count

' Dimension the related arrays

Am2 = rgA.Value

Am1 = rgA.Value

Ap1 = rgA.Value

Ap2 = rgA.Value

Fderiv = rgA.Value

' Input the function

Set rgF = Application.InputBox(Prompt:= _

  "The function is located in ", Type:=8)

rgF.Select

vF = rgF.Value

' Determine the partial derivatives

For i = 1 To rA

' Modify vA and read the corresponding F-values

  Am2(i, 1) = vA(i, 1) * (1 - 2 * Delta)

  rgA.Select

  Selection.Item(i, 1).Replace vA(i, 1), Am2(i, 1)

  rgF.Select

  Fm2 = rgF.Value

  Am1(i, 1) = vA(i, 1) * (1 - Delta)

  rgA.Select

  Selection.Item(i, 1).Replace Am2(i, 1), Am1(i, 1)

  rgF.Select

  Fm1 = rgF.Value

  Ap1(i, 1) = vA(i, 1) * (1 + Delta)

  rgA.Select

  Selection.Item(i, 1).Replace Am1(i, 1), Ap1(i, 1)

  rgF.Select

  Fp1 = rgF.Value

  Ap2(i, 1) = vA(i, 1) * (1 + 2 * Delta)

  rgA.Select

  Selection.Item(i, 1).Replace Ap1(i, 1), Ap2(i, 1)

  rgF.Select

  Fp2 = rgF.Value

  ' Restore A and compute the derivative

  rgA.Select

  Selection.Item(i, 1).Replace Ap2(i, 1), vA(i, 1)

  Fderiv(i, 1) = (Fm2 - Fp2 + 8 * (Fp1 - Fm1)) / _

    (12 * Delta * vA(i, 1))

Next i

' Write the results to the spreadsheet

rgF.Select

Selection.Offset(0, 1).Select

rgA.Select

Selection.Offset(0, 1).Select

Selection.Value = Fderiv

End Sub
'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub DecimalOperations()

' This program illustrates how to use the decimal format, 

' and what it can and cannot do.  

Dim vDec1, vDec2, vResult

' Test addition

vDec1 = "1000000000000.0000000000000234567"

vDec2 = "2000000000000.0000000000000456789"

Debug.Print "a = " & vDec1

Debug.Print "b = " & vDec2

vResult = CDec(vDec1) + CDec(vDec2)

MsgBox "a        = " & vDec1 & vbCr & _

  "b        = " & vDec2 & vbCr & _

  "a + b = " & vResult

Debug.Print ""

Debug.Print "Addition:"

' Test subtraction

vResult = CDec(vDec1) - CDec(vDec2)

Debug.Print "a + b = " & vResult   ' quadruple precision
Debug.Print vDec1 + vDec2          ' string concatenation
Debug.Print ""

MsgBox "a      = " & vDec1 & vbCr & _

  "b      = " & vDec2 & vbCr & _

  "a - b = " & vResult

Debug.Print "Subtraction:"

Debug.Print "a - b = " & vResult   ' quadruple precision
Debug.Print vDec1 - vDec2          ' double precision
Debug.Print ""

' Test multiplication

vResult = CDec(vDec1) * CDec(vDec2)

MsgBox "a       = " & vDec1 & vbCr & _

  "b       = " & vDec2 & vbCr & _

  "a * b = " & vResult

Debug.Print "Multiplication:"

Debug.Print "a * b = " & vResult   ' quadruple precision
Debug.Print vDec1 * vDec2          ' double precision
Debug.Print ""

' Test division

vResult = CDec(vDec1) / CDec(vDec2)

MsgBox "a      = " & vDec1 & vbCr & _

  "b      = " & vDec2 & vbCr & _

  "a / b = " & vResult

Debug.Print "Division:"

Debug.Print "a / b = " & vResult   ' quadruple precision
Debug.Print vDec1 / vDec2          ' double precision
Debug.Print ""

' Test exponentiation

vResult = Exp(decNum)

MsgBox "a      = " & vDec1 & vbCr & _

  "b      = " & vDec2 & vbCr & _

  "e^(a/b) = " & vResult

Debug.Print "Exponentiation:"

Debug.Print "e^(a/b) = " & vResult  ' double precision
Debug.Print Exp(vDec1 / vDec2)      ' double precision
Debug.Print ""

' Test taking the natural logarithm

vResult = CDec(vDec1) * Log(CDec(vDec2))

MsgBox "a         = " & vDec1 & vbCr & _

  "b         = " & vDec2 & vbCr & _

  "a ln(b) = " & vResult

Debug.Print "Taking the natural logarithm:"

Debug.Print "a ln(b) = " & vResult   ' quadruple precision
Debug.Print vDec1 * Log(vDec2)       ' double precision
Debug.Print ""

' Test taking the square root

vResult = Sqr(CDec(vDec1))

MsgBox "a           = " & vDec1 & vbCr & _

  "sqr(a)   = " & vResult

Debug.Print "Taking the square root:"

Debug.Print "sqr(a)  = " & vResult   ' double precision
Debug.Print Sqr(vDec1)               ' double precision
End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub CountingPennies()

' Decimal fractions, as occur in many currencies (dollars,

' euros, etc.) cannot be represented exactly in finite

' binary form. Consequently, additions of decimal fractions

' can lead to quite noticeable accumulation errors. For

' only a small number of additions, one can add them as

' integers, and subsequently move the decimal point back,

' but this trick only works as long as the resulting sum

' does not exceed the maximum size of 2,147,483,647 of

' "long" integers. In the example shown here, this approach

' will work for 6 million repeat additions of 1.23, but not

' for 7 million such additions. On Error Resume Next will

' keep the routine going even when LongSum overflows. (Note

' that its intervention will slow down the calculation. See

' what happens if you 'comment-out' this line.)

' Call this macro from the VBEditor (with F5) and see the 

' results appear in the Immediate Window, or call it from

' the spreadsheet and see the status bar display the value

' of n in millions.

' The accumulation error is clearly visible in DblSum, but

' not in DecSum.

Dim n As Long, LongSum As Long

Dim DblSum As Double

Dim DecSum As Variant

LongSum = 0

DblSum = 0

DecSum = 0

x = "3.21"

Debug.Print " n", " LongSum", " DblSum", _

  "               DecSum"

Debug.Print ""

Application.StatusBar = True           

For n = 1 To 10000000

  On Error Resume Next          ' to avoid overflow
  LongSum = LongSum + 321

  DblSum = DblSum + 3.21

  DecSum = DecSum + CDec(x)

  If n / 1000000 = CInt(n / 1000000) Then

    Debug.Print n, LongSum, DblSum, DecSum

    Application.StatusBar = "n = " & n 

  End If

Next n

Debug.Print ""

LongSum = LongSum / 100

Application.StatusBar = False          
End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'    MacroMorsels for spreadsheet and macro management    '
'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ScanSymbolCode()

' It may be desired to use a symbol available in MS Word, 

' but not readily available in Excel. If you use Excel as 

' part of MS Office, you can introduce those symbols into

' Excel after you figure out their character codenumber.

' First switch to Word, and select Insert -> Symbol. Under 

' the Symbols tab, in the Font window, scan the available

' fonts. Say that you look for a chemical equilibrium 

' sign, and find Wingdings 3. How do you find out what 

' its character codenumber is?

' Switch to Excel, and make a column starting in. e.g., 

' cell B4 with 0, 1, 2, etc., all the way down to, say, 39. 

' Place the cursor at the top of this column, then switch 

' to the VBEditor, and call the following MacroMorsel by 

' placing the cursor somewhere in its code, and then 

' pressing the function key F5. Keep that key down until 

' you have completed the column, then repeat the above 

' sequence in D4:D43 for 40 to 79, in F4:F43 with 80 etc.,  

' and finally in N4:N19 for 240 to 255.

' Now you have all symbols and their character codenumbers.  

' Label it in cell B2 with the symbol font name, perhaps 

' reduce the column widths to, say, 5.5, to make the 

' results fit a single page, then print it so that you have

' a hardcopy to keep for reference. Repeat for all fonts 

' that might be of interest, using the same lay-out. 

N = Selection.Value

Selection.Offset(0, 1).Select

Selection.Font.Name = "Wingdings 3"

Selection.Value = Chr(N)

Selection.Offset(1, -1).Select

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub UseSymbolCode()

' With ScanSymbolCode you found that the sign for chemical  

' equilibrium can be found in Wingdings 3 and there has the 

' character number code 68. To print this sysmbol in Excel,

' now use the following macro. Don’t forget the quotation

' marks around the font name, and the space before the 3.

Selection.Font.Name = "Wingdings 3"

Selection.Value = Chr(68)

' Say that you want to label a cell as A + B in equilibrium 

' with C + D. Here is how you can make that label. First

' use the present MacroMorsel to find the equilibrium sign. 

' Go to the formula bar; it will show there as another 

' symbol, such as a D. To its left type A + B (ending with 

' a space), and to its right C + D (starting with a space). 

' Then (still in the formula bar) highlight A + B and 

' change its font to Times New Roman, Arial, or whatever 

' standard text font you use, and repeat for C + D. 

' That’s it. 

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ControlANumericalDisplay()

' A common (and often valid) complaint of computer output

' is that it may display more digits than are statistically

' significant. This is mitigated by listing a corresponding

' imprecision measure, such as its standard deviation, but

' one might also want to limit the number of digits shown.

' When you round a number, the rounded digits are lost

' irretrievably. If the number may be needed in subsequent

' computations, that loss is often undesirable, as it may

' cause additional error. You may therefore want to control

' the number of digits displayed without actually rounding

' the number. This can be done by displaying the number in

' scientific notation with a set number of digits past the

' decimal point. You can then let the computer determine

' the desired number of digits to be displayed.

' Place a number containing many digits on the spreadsheet,

' highlight it, and call the macro. The number will be

' copied in the three cells to its immediate right, in

' scientific notation displayed with only two, three, or

' four digits, i.e., with one, two, or three digits past

' its decimal point. For comparison, a fourth cell contains

' the number rounded to one decimal place.

' By reformating these results back to a regular number

' format, or merely by looking at them in the formula box,

' verify that no digits were lost, except in the rounding.

' Experiment with the position of the decimal point in the

' input number, e.g., by using 123456789, 1234.56789, and

' 0.00123456789 respectively.

Dim vX, rndX

vX = Selection.Value             
 ' read the number
Selection.Offset(0, 1).Select

Selection.NumberFormat = "0.0E+00"   ' display the number
Selection.Value = vX

Selection.Offset(0, 1).Select

Selection.NumberFormat = "0.00E+00"  ' display the number
Selection.Value = vX

Selection.Offset(0, 1).Select

Selection.NumberFormat = "0.000E+00" ' display the number
Selection.Value = vX

Selection.Offset(0, 1).Select

rndX = Round(vX, 1)          

 ' round the number
Selection.Value = rndX

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub DeconstructACellAddress()

' The address of cell C consists of four parts: a dollar

' sign $, the column name (one or more letters), a second

' dollar sign, and the row number. The second dollar sign

' acts therefore as a convenient separator of column name

' cC and row number rC of cell C.

' To find the column name, find the location of the second

' dollar sign, then read the address string till there,

' deleting its left dollar sign.

' To find the row number, find the location of the second

' dollar sign, then read the string to the right of it.

' Highlight a spreadsheet cell, and call the macro. The

' column name and row number will be shown in the immediate

' window.

Dim aC, cC, rC, PosDollar2, CLeft

aC = Selection.Address

PosDollar2 = InStr(2, aC, "$")   ' yields position of 2nd $ sign in aC
CLeft = Left(aC, PosDollar2 - 1) ' yields string to left of PosDollar2
cC = Mid(CLeft, 2)               ' yields the column name
rC = Mid(aC, PosDollar2 + 1)     ' yields the row number
Debug.Print "Address = " & aC

Debug.Print "Column Name = " & cC & ",  Row Number = " & rC

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub DeconstructAnArrayAddress()

' In an array, i.e., a rectangular block of cells, a

' colon acts as the separator, and can therefore be

' used to isolate the top left address Atl from the

' bottom right address Abr. Those addresses can then

' be deconstructed separately. (Alternatively, one

' could use the 1st, 2nd, 3rd, and 4th dollar signs.)

' Highlight an array, then call this macro. The

' results will appear in the Immediate Window.

Dim aA, aAbr, aAtl, cAbr, cAtl, rAbr, rAtl

Dim ADollar, PosColon, PosDollar

aA = Selection.Address

PosColon = InStr(1, aA, ":")   ' yields the colon position
aAtl = Left(aA, PosColon - 1)  ' yields top left address
aAbr = Right(aA, PosColon - 1) ' yields bottom right address
PosDollar = InStr(2, aAtl, "$")

ADollar = Left(aAtl, PosDollar - 1)

cAtl = Mid(ADollar, 2)

rAtl = Mid(aAtl, PosDollar + 1)

PosDollar = InStr(2, aAbr, "$")

ADollar = Left(aAbr, PosDollar - 1)

cAbr = Mid(ADollar, 2)

rAbr = Mid(aAbr, PosDollar + 1)

Debug.Print "ArrayAddress:                " & aA

Debug.Print "TopLeft:      Cell Address = " & aAtl

Debug.Print "              Column Name  = " & cAtl

Debug.Print "              Row Number   = " & rAtl

Debug.Print "BottomRight:  Cell Address = " & aAbr

Debug.Print "              Column Name  = " & cAbr

Debug.Print "              Row Number   = " & rAbr

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ReconstituteAnEquation1()  

' This is a sampler of string manipulations to

' reconstitute a spreadsheet expression in VBA.

' For the sake of simplicity we will assume that

' the x-value is mentioned only once in F(x).

' Each step is documented. 

' Place a numerical value in a cell (preferably one

' between 0 and 0.5, for easy comparison with the

' table the macro will make), and a function using

' that value (once) in the cell to its right.

' Highlight both cells, then call this MacroMorsel.

' All intermediary and final results are displayed

' in the Immediate Window.

' Try it out for different locations, with a single or

' multiple-digit row, and/or single or double-letter

' columns.

Dim j As Single, LenF As Single, LenXX As Single

Dim PosDollar1 As Single, PosDollar2 As Single

Dim PosXinF as Single, PosXinF1 As Single

Dim vF As Double

Dim aX As String, ColumnStringX As String

Dim fF As String, RowStringX As String

Dim LeftF As String, RightF As String

Dim X As String, XX As String

' Read the input

aX = Selection(1, 1).Address    ' the address of x

fF = Selection(1, 2).Formula    ' the formula F(x)

' Optional additional testing

' You can quickly test alternative inputs scenarios by

' using a fake address and/or a fake function. To do this,

' make up your own values for aX, the address of x, and/or

' for the function F. An example is shown here; to use it,

' remove the hyphen in front of one or both of the following

' lines:

'aX = "$ABC$2345"

'fF = "=SQRT(1-ABC2345^2)"

' Find x

Debug.Print "aX = " & aX & ",   fF = " & fF

' Find the position of the first dollar sign

PosDollar1 = InStr(1, aX, "$")

Debug.Print "PosDollar1 = " & PosDollar1

' Find the position of the second dollar sign

PosDollar2 = InStr(2, aX, "$")

Debug.Print "PosDollar2 = " & PosDollar2

' Find the text between the 1st and 2nd dollar sign

ColumnStringX = Mid(aX, 2, PosDollar2 - PosDollar1 - 1)

Debug.Print "ColumnStringX = "; ColumnStringX

' Find the text to the right of the 2nd dollar sign

RowStringX = Mid(aX, PosDollar2 + 1, 8)

Debug.Print "RowStringX = " & RowStringX

' Combine the column name and row number

XX = ColumnStringX & RowStringX

Debug.Print "XX = " & XX

' Determine the length of the address of x

LenXX = Len(XX)

Debug.Print "LenXX = " & LenXX

Debug.Print ""

' The above ends the determination of the address of x.

' The next section finds the relevant parts of F(x).

' Start with printing the formula in the Immediate

' Window for easy reference:

' Determine the length of the address of x

LenF = Len(fF)

Debug.Print "LenF = " & LenF

' Determine the position, in the formula fF, of XX

PosXinF = InStr(1, fF, XX)

Debug.Print "PosXinF = " & PosXinF

' Find, in the formula fF, the text to the left of XX

LeftF = Left(fF, PosXinF - 1)

Debug.Print "LeftF = " & LeftF

' Find, in the formula fF, the text to the right of XX

RightF = Right(fF, LenF - PosXinF - LenXX + 1)

Debug.Print "RightF = " & RightF

Debug.Print ""

' These are all the pieces of the puzzle needed to

' reconstruct the expression. At the same time, we

' will check that it indeed works with some test

' values of X determined in the VBA code.

Debug.Print "Test that reconstituted formula works:"

For j = 0 To 5

  X = 0.1 * j

  vF = Application.Evaluate(LeftF & Str(X) & RightF)

  Debug.Print "X = " & X & ",   vF = " & vF

Next j

Debug.Print ""

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ReconstituteAnEquation2()

' This is a sampler of string manipulations to

' reconstitute a spreadsheet expression F(x) of

' a single variable x in VBA. The x-value can be

' mentioned multiple times (here arbitrarily set to

' maximally six times) in the expression for F(x).

' Place a numerical value in a cell, and a function

' using that value in the cell to its right.

' Highlight both cells, then call this MacroMorsel.

' All intermediary and final results are displayed

' in the Immediate Window.

' Try it out for different locations, with a single

' or multiple-letter column name, and/or a single

' or multiple-digit row number.

' A few suggested trial functions:

'   y=exp(x^2)*erfc(x)

'   y=x^3-2*x^2+3x-4

'   y = (exp(2*x)-exp(-2*x))/(exp(2*x)+exp(-2*x))

Dim i As Single, iMax As Single, j As Single

Dim LenF As Single, LenXX As Single, PosDollar1 As Single

Dim PosDollar2 As Single, PosXinF As Single

Dim vF As Double

Dim aX As String, ColumnStringX As String, fF As String

Dim LeftF(1 To 20) As String, LeftFF As String

Dim RightF(1 To 20) As String, RowStringX As String

Dim X As String, XX As String

i = 1

' Read the input

aX = Selection(1, 1).Address

fF = Selection(1, 2).Formula

' Optional additional testing

' You can quickly test alternative inputs scenarios by using

' a fake address and/or a fake function. To do this, make up

' your own values for the address of x, aX, and/or for the

' function F. An example is shown here; to use it, remove

' the hyphen in front of one or both of the following lines:

'aX = "$ABC$2345"

'fF = "=3*ABC2345^3-2*ABC2345^2+ABC2345-1"

' Find x

Debug.Print "aX = " & aX

PosDollar1 = InStr(1, aX, "$")

Debug.Print "PosDollar1 = " & PosDollar1

PosDollar2 = InStr(2, aX, "$")

Debug.Print "PosDollar2 = " & PosDollar2

ColumnStringX = Mid(aX, 2, _

  PosDollar2 - PosDollar1 - 1)

Debug.Print "ColumnStringX = "; ColumnStringX

RowStringX = Mid(aX, PosDollar2 + 1, 8)

Debug.Print "RowStringX = " & RowStringX

XX = ColumnStringX & RowStringX

Debug.Print "XX = " & XX

LenXX = Len(XX)

Debug.Print "LenXX = " & LenXX

Debug.Print ""

' Find the relevant parts of F(x)

Debug.Print "fF = " & fF

LenF = Len(fF)

Debug.Print "LenF = " & LenF

PosXinF = InStr(1, fF, XX)

Debug.Print "PosXinF = " & PosXinF

LeftF(i) = Left(fF, PosXinF - 1)

Debug.Print "LeftF(" & i & ") = " & LeftF(i)

RightF(i) = Right(fF, LenF - PosXinF - LenXX + 1)

Debug.Print "RightF(" & i & ") = " & RightF(i)

Debug.Print "i = " & i

Debug.Print ""

' Check whether RightF(1) contains more terms XX

PosXinF = InStr(1, RightF(1), XX)

Debug.Print PosXinF

' If RightF(i) still contains XX, continue parsing

Do

  If PosXinF = 0 Then GoTo A:

  Debug.Print "RightF(" & i & ") = " & RightF(i)

  LenF = Len(RightF(i))

  Debug.Print "LenF = " & LenF

  PosXinF = InStr(1, RightF(i), XX)

  Debug.Print "PosXinF = " & PosXinF

  If PosXinF <= 0 Then GoTo A

  LeftF(i + 1) = Left(RightF(i), PosXinF - 1)

  Debug.Print "LeftF(" & i + 1 & ") = " & LeftF(i + 1)

  RightF(i + 1) = Right(RightF(i), _

    LenF - PosXinF - LenXX + 1)

  Debug.Print "RightF(" & i + 1 & ") = " & RightF(i + 1)

  i = i + 1

  Debug.Print "i = " & i

  Debug.Print ""

  PosXinF = InStr(1, RightF(i), XX)

Loop Until PosXinF <= 0

A:

' Reconstitute and test the expression

iMax = i

  Debug.Print "iMax = " & iMax

  Debug.Print ""

For j = 1 To 5

  X = 0.1 * j

If iMax = 1 Then vF = Application.Evaluate _

  (LeftF(1) & Str(X) & RightF(1))

If iMax = 2 Then vF = Application.Evaluate _

  (LeftF(1) & Str(X) & LeftF(2) & Str(X) & RightF(2))

If iMax = 3 Then vF = Application.Evaluate _

  (LeftF(1) & Str(X) & LeftF(2) & Str(X) & _

  LeftF(3) & Str(X) & RightF(3))

If iMax = 4 Then vF = Application.Evaluate _

  (LeftF(1) & Str(X) & LeftF(2) & Str(X) & _

  LeftF(3) & Str(X) & LeftF(4) & Str(X) & RightF(4))

If iMax = 5 Then vF = Application.Evaluate _

  (LeftF(1) & Str(X) & LeftF(2) & Str(X) & _

  LeftF(3) & Str(X) & LeftF(4) & Str(X) & _

  LeftF(5) & Str(X) & RightF(5))

If iMax = 6 Then vF = Application.Evaluate _

  (LeftF(1) & Str(X) & LeftF(2) & Str(X) & _

  LeftF(3) & Str(X) & LeftF(4) & Str(X) & _

  LeftF(5) & Str(X) & LeftF(6) & Str(X) & RightF(6))

If iMax > 6 Then

  MsgBox "The variable x appears more than" & _

    Chr(10) & "six times in the expression for F(x)." & _

    Chr(10) & "Please adjust the macro accordingly."

  End

End If

  Debug.Print "F(" & X & ") = " & vF

Next j

Debug.Print ""

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub MoveAndResizeARange()

' This MacroMorsel demonstrates how you can move and

' resize a range, by combining Offset and Resize.

' In cells A1:A10 deposit the numbers 1 (1) 10. In cell B1

' deposit the instruction =A1+10, copy this instruction

' to C1:G1, then copy B1:G1 down to row 10. Block A1:G10

' will form our test data set.

' Call the macro, in the input box enter a small

' subsection of this block, e.g., B2:C4, and push OK.

' Look in the Immediate Window and see what you get.

' You are only reading the ranges, not trying to move

' them. Then try some other input blocks,other offsets,

' and other final sizes.

' You can read the formulas instead of the values by

' everywhere specifying fY = Selection.Formula instead

' of vY = Selection.Value, and replacing vY by fY.

' For a long list of output values, you may want to move

' the Immediate Window next to (rather than underneath)

' the module (by dragging its title bar), and extend its

' length (by dragging its bottom edge down when you see

' its two-headed arrow).

' Please note that the input range must be at least two

' rows high, otherwise the resizing will yield an error.

Dim c As Integer, cMax As Integer

Dim r As Integer, rMax As Integer

Dim vY As Variant

Dim aY As String

Dim rgY As Range

' Read the range rgY, then read its values into array Y

Set rgY = Application.InputBox(prompt:="rgY", Type:=8)

rgY.Select

rMax = Selection.Rows.Count

cMax = Selection.Columns.Count

vY = Selection.Value    ' This dimensions the array

' Display the array Y in the Immediate Window. Note how

' the instruction combines text (within quotation marks)

' and index numbers (r and c), linked by ampersands.

Debug.Print "Before offset & reset:"

Debug.Print ""

Debug.Print "aY = " & Selection.Address

Debug.Print ""

For c = 1 To cMax

  For r = 1 To rMax

    Debug.Print "vY(" & r & "," & c & ") = " & vY(r, c)

  Next r

Next c

' Redefine the new range size

cMax = cMax * 2

rMax = rMax - 1

' Resize the range, and reread its values

rgY.Select

Selection.Offset(2, 1).Select

Selection.Resize(rMax, cMax).Select

vY = Selection.Value    ' This redimensions the array

' Display the new range in the Immediate Window

Debug.Print ""

Debug.Print "After offset & reset:"

Debug.Print ""

Debug.Print "aY = " & Selection.Address

Debug.Print ""

For c = 1 To cMax

  For r = 1 To rMax

    Debug.Print "vY(" & r & "," & c & ") = " & vY(r, c)

  Next r

Next c

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub InhibitScreenUpdating()

' This MacroMorsel uses timing (as described in Advanced  

' Excel section 8.10) to demonstrate the speed difference

' obtainable by inhibiting screen updating.

' Highlight an empty cell before calling this macro;

' the cell immediately below it should also be empty.

' Then call the macro. You can enhance the effect by

' highlighting a cell block rather than a single cell.

' You need not "clean up" afterwards, because screen

' updating is automatically reset to True just before the

' subroutine ends, so that it does not affect its final 

' output.

' Preferably inhibit screen updating after the macro has

' extracted all needed information from the spreadsheet,

' because inhibiting screen updating also prohibits 

' drag-and-drop, but before it starts computations that

' might display intermediate results on the monitor screen. 

Dim i As Integer, j As Integer

Dim Time1, Time2, seconds

Time1 = Timer

For i = 1 To 5000

  j = i

  Selection.Value = j

Next i

Time2 = Timer

Debug.Print "With screen updating, Interval = " & Time2 - Time1 & " seconds"

Selection.Offset(1, 0).Select

Application.ScreenUpdating = False

Time1 = Timer

For i = 1 To 1500

  j = i

  Selection.Value = j

Next i

Time2 = Timer

Debug.Print "Without screen updating, Interval = _

  " & Time2 - Time1 & " seconds"

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub KeepUserInformed()

' This MacroMorsel shows intermediate status information

' on the Status Bar (in the bottom left of the screen)

' about the progress of a slow calculation (here: counting

' from 1 to 40 million).

Dim i As Long, j As Double

Application.StatusBar = True           ' turn StatusBar on
Application.StatusBar = "The process is 0 % complete."

For i = 1 To 40000000

  j = i

  If Int(j / 2000000) = j / 2000000 Then _

    Application.StatusBar = "The process is " _

    & j / 400000 & " % complete."      'here: in 5% steps

Next i

MsgBox "You just counted to " & j

Application.StatusBar = False          'clean up
End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub AttachCellComments()

' Highlight a cell before calling this macro. In a

' realistic application, the cell will most likely be

' specified differently.

' See what difference it makes to specify Comment.Visible

' as True. You can get the same effect on the spreadsheet

' by right-clicking on the commented cell and selecting

' Show/Hide Comments, a toggle.

' Also play with the Shape.Width and Shape.Height; you

' can delete these instructions if you want the default-

' size comment box.

' Cell Comments can be used to identify the input range of

' an array operation, the name of the macro used, the date

' and time of use (which the computer will provide), etc.,

' information that may not be available otherwise.

ActiveCell.Select

ActiveCell.ClearComments        ' clear prior comments
ActiveCell.AddComment

ActiveCell.Comment.Visible = False
ActiveCell.Comment.Shape.Width = 135      ' default 96
ActiveCell.Comment.Shape.Height = 60      ' default 56
ActiveCell.Comment.Text Text:="Whatever text and/or" _

  & Chr(10) & "information you may wish" & Chr(10) & _

  "to make avalable here" & Chr(10) & Chr(10) & _

  "Date: " & Date & ", Time: " & Time

' Notes:

'   (1) Comment.Visible = True makes the comment permanently

' visible; Comment.Visible = False makes it visible only

' while the pointer hovers over it. In both cases the top

' right cell corner shows a red triangle.

'   (2) Double-clicking on the cell yields tools such as

' Edit Comment, Delete Comment, and the toggle switch

' between Hide Comment and Show/Hide Comments.

'   (2) You need not specify the width and/or height of the 

' comment box; they will then assume the default values of

' about 96 for its width, about 56 for its height.

'   (4) Use Chr(10) for carriage return;

' vbCr doesn't work in a Comment.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub MakeTestToolbar()

' It is often convenient to have quick access to a set of

' macros of your own choice. A toolbar allows you this.

' Here is a sampler to show how you can make macros A

' through I easily available from the spreadsheet. Their

' farciful names are of course immaterial, and the toolbar

' will only work when there are real macros to back it up.

' Otherwise, clicking on the macro buttons will show a

' message box will informing you that the requested macro

' cannot be found.

' Button1 (or any other name) will give access to a single

' macro, Button2 to a choice of two, and Button3 to three

' submenus, of which two have further options. Using four

' or more levels of nesting is seldom convenient.

' A Popup is an intermediary stage, used to specify a

' group of macros; you must use a Button to call a macro.

' An OnAction instruction then calls that particular macro.

' The command BeginGroup controls a vertical line to the

' left of a Popup to visually separate distinct categories.

' The other command names are rather self-explanatory.

Dim TestToolBar As CommandBar

Dim Button1 As CommandBarButton

Dim Button2 As CommandBarPopup

Dim Button21 As CommandBarButton

Dim Button22 As CommandBarButton

Dim Button3 As CommandBarPopup

Dim Button31 As CommandBarButton

Dim Button32 As CommandBarPopup

Dim Button321 As CommandBarButton

Dim Button322 As CommandBarButton

Dim Button323 As CommandBarButton

Dim Button33 As CommandBarPopup

Dim Button331 As CommandBarButton

Dim Button332 As CommandBarButton

' Delete earlier version of TestToolBar,

' if existing, to prevent conflicts

On Error Resume Next

CommandBars("TestToolBar").Delete

' Create a commandbar

Set TestToolBar = CommandBars.Add

With TestToolBar

  .Name = "TestToolBar"

  .Position = msoBarTop

  .Visible = True

End With

' Create Button1 for macro A

Set Button1 = CommandBars("TestToolbar") _

  .Controls.Add(Type:=msoControlButton)

With Button1

  .Caption = "Appreciate"

  .Style = msoButtonCaption

  .OnAction = "MacroA"

End With

' Create a submenu and buttons for macros B & C

Set Button2 = CommandBars("TestToolbar") _

  .Controls.Add(Type:=msoControlPopup)

Button2.Caption = "Behold"

Button2.BeginGroup = True

Set Button21 = Button2.Controls _

  .Add(Type:=msoControlButton)

With Button21

  .Caption = "Binary"

  .Style = msoButtonCaption

  .OnAction = "MacroB"

End With

Set Button22 = Button2.Controls _

  .Add(Type:=msoControlButton)

With Button22

  .Caption = "Complement"

  .Style = msoButtonCaption

  .OnAction = "MacroC"

End With

' Create a menu for macros D through I

Set Button3 = CommandBars("TestToolbar") _

  .Controls.Add(Type:=msoControlPopup)

With Button3

  .Caption = "Contemplate"

  .BeginGroup = True

End With

' Create a button for macro D 

Set Button31 = Button3.Controls _

  .Add(Type:=msoControlButton)

With Button31

  .Caption = "Divide"

  .Style = msoButtonCaption

  .OnAction = "MacroD"

End With

' Create a submenu and buttons for macros E, F, and G

Set Button32 = Button3.Controls.Add(Type:=msoControlPopup)

Button32.Caption = "EFG"

Set Button321 = Button32.Controls _

  .Add(Type:=msoControlButton)

With Button321

  .Caption = "Exponentiate"

  .Style = msoButtonCaption

  .OnAction = "MacroE"

End With

Set Button322 = Button32.Controls _

  .Add(Type:=msoControlButton)

With Button322

  .Caption = "Function"

  .Style = msoButtonCaption

  .OnAction = "MacroF"

End With

Set Button323 = Button32.Controls _

  .Add(Type:=msoControlButton)

With Button323

  .Caption = "Good Luck"

  .Style = msoButtonCaption

  .OnAction = "MacroG"

End With

' Create a submenu and buttons for macros H and I

Set Button33 = Button3.Controls _

  .Add(Type:=msoControlPopup)

Button33.Caption = "H+I"

Set Button331 = Button33.Controls _

  .Add(Type:=msoControlButton)

With Button331

  .Caption = "Highlight"

  .Style = msoButtonCaption

  .OnAction = "MacroH"

End With

Set Button332 = Button33.Controls _

  .Add(Type:=msoControlButton)

With Button332

  .Caption = "Integrate"

  .Style = msoButtonCaption

  .OnAction = "MacroI"

End With

' You can add embellishments, such as .TooltipText =

' "your text", to the top-tier toolbar buttons.

' For a worked-out example see the InsertMBToolbar

' macro in the MacroBundle.

' Note that Excel 2007 only allows a toolbar in its

' Developer ribbon.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub DeleteTestToolbar()

On Error Resume Next

CommandBars("TestToolBar").Delete

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ErrorPrevention()

' The possibility of certain types of errors can be

' forseen, in which case they can be prevented by proper

' coding. A common type of error is when a macro calls

' a function to perform an "illegal" operation, such as

' taking the square root of a negative number, or the

' logarithm of a non-positive argument, or divides by zero.

' Here is a simple macro that reads a value from the spread-

' sheet, computes its square root and its inverse, and then

' writes these back onto the spreadsheet. Try it with a

' positive input number, then with 0, and finally with a

' negative number. Colorcoding is used to alert the user

' that a valid answer could not be obtained.

Dim n As Double

Dim B, C, n

n = Selection.Value

If n < 0 Then

  B = "n < 0"

  GoTo P

End If

B = Sqr(n)

P:

If n = 0 Then

  C = "n = 0"

  GoTo Q

End If

C = 1 / n

Q:

Selection.Offset(0, 1).Select

Selection.Value = B

ActiveCell.Font.ColorIndex = 1

If n < 0 Then ActiveCell.Font.ColorIndex = 3

Selection.Offset(0, 1).Select

Selection.Value = C

ActiveCell.Font.ColorIndex = 1

If n = 0 Then ActiveCell.Font.ColorIndex = 3

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ErrorHandling1()

' Forseeable errors can also be avoided by proper error

' handling, although this is usually a less elegant

' solution than preventing the problem in the first place.

' Here is a simple macro that reads a value from the

' spreadsheet, and takes its square root. As in the

' ErrorPrevention MacroMorsel, try it with a positive

' number, then with 0, and finally with a negative number.

Dim n As Double

Dim B

On Error GoTo ErrorHandler

n = Selection.Value

B = Sqr(n)

P:

Selection.Offset(0, 1).Select

Selection.Value = B

Exit Sub

ErrorHandler:

  If Err.Number = 5 Then

    MsgBox "  The argument of the" & Chr(13) _

      & "square root is negative."

    B = "N/A"

  End If

  Resume P

' ErrorHandling2 shows you how to find the error number.

' While the error can occur anywhere in a routine, it is

' usually best to place the error handler(s) at the end

' of that routine.

' Exit an error handler with either Resume or Exit Sub

' (or Exit Function). The Resume instruction followed by

' a label resumes code execution at the label A, allowing

' the program to continue execution if so desired.

' By handling specific errors that you can anticipate, you

' do not mask messages that point to other errors, as you

' would with a general error-avoidance instruction such as

' On Error Resume Next, which merely hides all problems.

' Note that the error handler cannot be used with another

' common type of error, which occurs when a macro calls a

' non-functional routine, such as when one tries to run

' SolverScan without having properly installed the special

' Solver functions, or tries to implement an xnumbers macro

' without having loaded the required xnumbers.dll.

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ErrorHandling2()

' The following shows how to use error trapping to find 

' the specific error number.

Dim n As Double

Dim B

On Error GoTo ErrorHandler

n = Selection.Value

B = Sqr(n)

Selection.Offset(0, 1).Select

Selection.Value = B

P:

Exit Sub

ErrorHandler:

  MsgBox Err.Number

  Resume P

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

